Maple-assisted proof of formula for A207724

Robert Israel

5 March 2018

Of the $2^6 = 64$ possible configurations for a 2×3 sub-array, 36 avoid 0 0 0 and 0 1 0 horizontally. Consider the 36×36 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×3 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j), and 0 otherwise. The following Maple code computes it.

```
> Configs:= select(A -> A[1..3] <> [0,0,0] and A[4..6] <> [0,0,0]
  and A[1..3] <> [0,1,0] and A[4..6] <> [0,1,0],
  [seq(convert(x,base,2)[1..6],x=2^6..2^7-1)]): nops(Configs);
```

$$36 \quad (1)$$

```
> compat:= proc(i,j) local k,col;
     if Configs[i][4..6] <> Configs[j][1..3] then return 0 fi;
     for k from 1 to 3 do
        col:= [Configs[i][k],Configs[i][k+3],Configs[j][k+3]];
        if col = [0,1,1] or col = [1,0,1] then return 0 fi;
     od;
     1
  end proc:
  T:= Matrix(36,36,compat);
```

$$T := \begin{bmatrix} \text{36 x 36 Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix} \quad (2)$$

Thus for $n \geq 2$, $a(n) = u\, T^{n-2}\, v$ where u and v are 36-dimensional row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](36, 1):
  v:= Vector(36,1):
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^(n-2) . v,n=2..12);
```

$$36, 78, 189, 490, 1113, 2449, 5474, 12036, 26100, 56595, 122472 \quad (3)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{11} - 3\,t^{10} + 2\,t^9 - 3\,t^8 + 6\,t^7 - 3\,t^4 - t^3 + t \quad (4)$$

This turns out to have degree 11 but no constant term. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=1}^{11} p_i\, a(i+n+2)$ for $n \geq 0$, where p_i is the coefficient of t^i in $P(t)$. This is a homogeneous linear recurrence of order 10.

```
> rec:= add(coeff(P(t), t, i)*a(i+n+2),i=1..11)=0;
```

$$rec := a(3+n) - a(5+n) - 3\,a(6+n) + 6\,a(9+n) - 3\,a(10+n) + 2\,a(11+n) - 3\,a(12+n) + a(13+n) = 0 \quad (5)$$

The "empirical" formula is a shifted version of this. It is easily verified that the formula above is also valid for $n = -2$ and $n = -1$.